Park City Math Institute ~ Institute for Advanced Study

An Introduction to DifferentialGeometry with Maple
PCMI - July 5, 2013

The Eguchi-Hanson metric

Synopsis

The Eguchi–Hanson metric is an example of a gravitational instanton (complete, Riemannian, Ricci-flat). In this worksheet we shall:

- define the Eguchi-Hanson metric and build an orthonormal frame for calculating its properties
- show that the metric is Ricci-flat and self-dual.
- show that the holonomy is so(3)
- explicitly prove that the metric admits a quaternionic-Kahler structure.
- calculate the Kahler potential for the metric

The Eguchi-Hanson metric

```
M > restart:
> with(DifferentialGeometry): with(Tensor): with(LieAlgebras):
```

Define the coordinates with DGsetup.

```
> DGsetup([R, theta, phi, psi], M0);
```

$$\textit{frame name: M0} \qquad (1.1)$$

Define a set of three1-forms used to construct an orthogonal frame for the metric.

```
> s1 := evalDG(sin(psi)*dtheta - cos(psi)*sin(theta) *dphi);
```

$$s1 := \sin(\psi)\, d\theta - \cos(\psi)\,\sin(\theta)\, d\phi \qquad (1.2)$$

```
> s2 := evalDG(cos(psi)*dtheta + sin(psi)*sin(theta)*dphi);
```

$$s2 := \cos(\psi)\, d\theta + \sin(\psi)\,\sin(\theta)\, d\phi \qquad (1.3)$$

```
> s3 := evalDG(dpsi + cos(theta)*dphi);
```

$$s3 := \cos(\theta)\, d\phi + d\psi \qquad (1.4)$$

Define an anholomonic frame on M0. We shall do all our calculations with respect to this frame.

```
M > FD := FrameData([dR, s1, s2, s3], M);
```

$$FD := \left[d\,\Theta1 = 0, d\,\Theta2 = -\Theta3 \wedge \Theta4, d\,\Theta3 = \Theta2 \wedge \Theta4, d\,\Theta4 = -\Theta2 \wedge \Theta3\right] \qquad (1.5)$$

Initialize this frame with DGsetup.

```
M > DGsetup(FD, [zeta0, zeta1, zeta2, zeta3], [sigma0, sigma1, sigma2, sigma3]);
```

$$\text{frame name: } M \tag{1.6}$$

Here is the Eguchi-Hansen metric in this frame

```
M > g := evalDG((1 - a/R^4)^(-1)*sigma0 &t sigma0 + R^2/4*(1 - a/R^4)*sigma3 &t sigma3 + R^2/4*(sigma1 &t sigma1 + sigma2 &t sigma2));
```

$$g := \frac{R^4}{R^4 - a}\,\sigma 0 \otimes \sigma 0 + \frac{R^2}{4}\,\sigma 1 \otimes \sigma 1 + \frac{R^2}{4}\,\sigma 2 \otimes \sigma 2 + \frac{R^4 - a}{4\,R^2}\,\sigma 3 \otimes \sigma 3 \tag{1.7}$$

Curvature

The Eguchi-Hanson metric is Ricci-flat and self-dual.

```
M > RicciTensor(g);
```

$$0\,\sigma 0 \otimes \sigma 0 \tag{2.1}$$

We calculate the curvature tensor and its dual and check that these are equal.

```
M > C := CurvatureTensor(g):
M > C1 := RaiseLowerIndices(g, C, [1]):
M > Cdual := DualCurvature(g, C) assuming R > 0:
M > C1 &minus Cdual;
```

$$0\,\sigma 0 \otimes \sigma 0 \otimes \sigma 0 \otimes \sigma 0 \tag{2.2}$$

Infinitesimal Holonomy

The infinitesimal holonomy is a Lie algebra of matrices which act on the tangent space of the manifold at a given point by isometries.

```
M > H := InfinitesimalHolonomy(g, [R = a, theta = Pi/2, phi = 0 , psi = 0]);
```

$$H := \left[\begin{bmatrix} 0 & -\frac{a}{2\,R^4} & 0 & 0 \\ \frac{2\,a}{\left(R^4 - a\right)\,R^2} & 0 & 0 & 0 \\ 0 & 0 & 0 & -\frac{a}{R^5} \\ 0 & 0 & \frac{a}{\left(R^4 - a\right)\,R} & 0 \end{bmatrix}, \right. \tag{3.1}$$

$$\left[\begin{bmatrix} 0 & 0 & -\frac{a}{2R^4} & 0 \\ 0 & 0 & 0 & \frac{a}{R^5} \\ \frac{2a}{(R^4-a)R^2} & 0 & 0 & 0 \\ 0 & -\frac{a}{(R^4-a)R} & 0 & 0 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & \frac{a(R^4-a)}{R^8} \\ 0 & 0 & \frac{2a}{R^5} & 0 \\ 0 & -\frac{2a}{R^5} & 0 & 0 \\ -\frac{4a}{(R^4-a)R^2} & 0 & 0 & 0 \end{bmatrix}\right]$$

Here are the structure equations for the holonomy algebra.

```
M > LD := map(simplify, LieAlgebraData(H, hol));
```

$$LD := \left[[e1, e2] = -\frac{a\,e3}{(R^4-a)\,R}, [e1, e3] = \frac{4\,a\,e2}{R^5}, [e2, e3] = -\frac{4\,a\,e1}{R^5}\right] \quad (3.2)$$

We can simplify these structure equations by scaling the vectors.

```
M > DGsetup(LD);
```

$$\textit{Lie algebra: hol} \quad (3.3)$$

```
hol > LD2  := LieAlgebraData([r*e1, s*e2, t*e3], alg);
```

$$LD2 := \left[[e1, e2] = -\frac{r\,s\,a\,e3}{t\,(R^4-a)\,R}, [e1, e3] = \frac{4\,r\,t\,a\,e2}{s\,R^5}, [e2, e3] = -\frac{4\,s\,t\,a\,e1}{r\,R^5}\right] \quad (3.4)$$

```
hol > S := solve(subs({-r*s*a/(t*(R^4-a)*R) = 1, 4*r*t*a/(s*
      R^5) = -1, -4*s*t*a/(r*R^5) = 1}), {r,s,t}, explicit);
```

$$S := \left\{r = \frac{\sqrt{R^4-a}\,R^3}{2\,a}, s = \frac{\sqrt{R^4-a}\,R^3}{2\,a}, t = -\frac{R^5}{4\,a}\right\}, \left\{r = -\frac{\sqrt{R^4-a}\,R^3}{2\,a}, s = -\frac{\sqrt{R^4-a}\,R^3}{2\,a}, t = -\frac{R^5}{4\,a}\right\}, \left\{r = \frac{\sqrt{R^4-a}\,R^3}{2\,a}, s = -\frac{\sqrt{R^4-a}\,R^3}{2\,a}, t = \frac{R^5}{4\,a}\right\}, \left\{r = -\frac{\sqrt{R^4-a}\,R^3}{2\,a}, s = \frac{\sqrt{R^4-a}\,R^3}{2\,a}, t = \frac{R^5}{4\,a}\right\} \quad (3.5)$$

```
hol > subs(S[1], LD2);
```
$$[[e1, e2] = e3, [e1, e3] = -e2, [e2, e3] = e1] \quad (3.6)$$

We conclude that our metric has infinitesimal holonomy given by so(3).

Quaternionic-Kahler Structure

A quaternionic- Kahler structure consists of three covariantly constant $\binom{1}{1}$ tensors J_1, J_2, J_3, which satisfy the algebraic relations of the quaternions. Because these tensors are covariantly constant and square to -1, each define a Kahler structure.

It is possible to calculate the tensors J_1, J_2, J_3 directly with the command CovariantlyConstantTensors but the calculation runs much faster if we first calculate the 1 -1 tensors which are point-wise invariant with respect to the holonomy matrices.

First, generate all the 1-1 tensors.

```
hol > OneOneTensors := GenerateTensors([[zeta0, zeta1, zeta2,
      zeta3], [sigma0, sigma1, sigma2, sigma3]]);
```
$$OneOneTensors := [\zeta0 \otimes \sigma0, \zeta0 \otimes \sigma1, \zeta0 \otimes \sigma2, \zeta0 \otimes \sigma3, \zeta1 \otimes \sigma0, \zeta1 \otimes \sigma1, \zeta1 \otimes \sigma2, \zeta1 \otimes \sigma3, \zeta2 \otimes \sigma0, \zeta2 \otimes \sigma1, \zeta2 \otimes \sigma2, \zeta2 \otimes \sigma3, \zeta3 \otimes \sigma0, \zeta3 \otimes \sigma1, \zeta3 \otimes \sigma2, \zeta3 \otimes \sigma3] \quad (4.1)$$

Find the 1-1 tensors which are invariant with respect to the holonomy.

```
M > HolInvTensors := InvariantTensorsAtAPoint(H, OneOneTensors);
```
$$HolInvTensors := \left[\zeta0 \otimes \sigma0 + \zeta1 \otimes \sigma1 + \zeta2 \otimes \sigma2 + \zeta3 \otimes \sigma3, \frac{R^4 - a}{2 R^3} \zeta0 \otimes \sigma1 - \frac{2}{R} \zeta1 \otimes \sigma0 - \frac{R^4 - a}{R^4} \zeta2 \otimes \sigma3 + \zeta3 \otimes \sigma2, -\frac{R^4 - a}{2 R^3} \zeta0 \otimes \sigma2 - \frac{R^4 - a}{R^4} \zeta1 \otimes \sigma3 + \frac{2}{R} \zeta2 \otimes \sigma0 + \zeta3 \otimes \sigma1, -\frac{(R^4 - a)^2}{4 R^6} \zeta0 \otimes \sigma3 + \frac{R^4 - a}{2 R^3} \zeta1 \otimes \sigma2 - \frac{R^4 - a}{2 R^3} \zeta2 \otimes \sigma1 + \zeta3 \otimes \sigma0\right] \quad (4.2)$$

Find the covariantly constant tensors in the span of the holonomy invariant tensors.

```
M > C := Christoffel(g):
M > CCOneOneTensors := CovariantlyConstantTensors(C,
    HolInvTensors[2 .. 4]);
```
(4.3)

$$CCOneOneTensors := \left[-\frac{R^4-a}{4R^3}\, \zeta 0 \otimes \sigma 3 + \frac{1}{2}\, \zeta 1 \otimes \sigma 2 - \frac{1}{2}\, \zeta 2 \otimes \sigma 1 + \frac{R^3}{R^4-a}\, \zeta 3 \otimes \sigma 0, -\frac{\sqrt{R^4-a}}{2R}\, \zeta 0 \otimes \sigma 2 - \frac{\sqrt{R^4-a}}{R^2}\, \zeta 1 \otimes \sigma 3 + \frac{2R}{\sqrt{R^4-a}}\, \zeta 2 \otimes \sigma 0 + \frac{R^2}{\sqrt{R^4-a}}\, \zeta 3 \otimes \sigma 1, \frac{\sqrt{R^4-a}}{2R}\, \zeta 0 \otimes \sigma 1 - \frac{2R}{\sqrt{R^4-a}}\, \zeta 1 \otimes \sigma 0 - \frac{\sqrt{R^4-a}}{R^2}\, \zeta 2 \otimes \sigma 3 + \frac{R^2}{\sqrt{R^4-a}}\, \zeta 3 \otimes \sigma 2 \right] \tag{4.3}$$

Now scale the results to get tensors with the correct algebraic properties.

```
M > J0 := evalDG(zeta0 &t sigma0 + zeta1 &t sigma1 + zeta2 &t
    sigma2 + zeta3 &t sigma3);
```

$$J0 := \zeta 0 \otimes \sigma 0 + \zeta 1 \otimes \sigma 1 + \zeta 2 \otimes \sigma 2 + \zeta 3 \otimes \sigma 3 \tag{4.4}$$

```
M > J1 := 2 &mult evalDG(-(R^4 - a)/(4*R^3)*zeta0 &t sigma3 +
    1/2*zeta1 &t sigma2 -1/2*zeta2 &t sigma1 + R^3/(R^4 - a)*
    zeta3 &t sigma0);
```

$$J1 := -\frac{R^4-a}{2R^3}\, \zeta 0 \otimes \sigma 3 + \zeta 1 \otimes \sigma 2 - \zeta 2 \otimes \sigma 1 + \frac{2R^3}{R^4-a}\, \zeta 3 \otimes \sigma 0 \tag{4.5}$$

```
M > J2 := (-1) &mult evalDG(-sqrt(R^4 - a)/(2*R)*zeta0 &t sigma2
    - sqrt(R^4 -a)/(R^2)*zeta1 &t sigma3 + 2*R/sqrt(R^4 -a)*zeta2
    &t sigma0 + R^2/sqrt(R^4 -a)*zeta3 &t sigma1);
```

$$J2 := \frac{\sqrt{R^4-a}}{2R}\, \zeta 0 \otimes \sigma 2 + \frac{\sqrt{R^4-a}}{R^2}\, \zeta 1 \otimes \sigma 3 - \frac{2R}{\sqrt{R^4-a}}\, \zeta 2 \otimes \sigma 0 - \frac{R^2}{\sqrt{R^4-a}}\, \zeta 3 \otimes \sigma 1 \tag{4.6}$$

```
M > J3 := evalDG(sqrt(R^4 - a)/(2*R)*zeta0 &t sigma1 -2*R/sqrt
    (R^4 - a)*zeta1 &t sigma0 -sqrt(R^4-a)/R^2 *zeta2 &t sigma3 +
    R^2/sqrt(R^4 - a)* zeta3 &t sigma2);
```

$$J3 := \frac{\sqrt{R^4-a}}{2R}\, \zeta 0 \otimes \sigma 1 - \frac{2R}{\sqrt{R^4-a}}\, \zeta 1 \otimes \sigma 0 - \frac{\sqrt{R^4-a}}{R^2}\, \zeta 2 \otimes \sigma 3 + \frac{R^2}{\sqrt{R^4-a}}\, \zeta 3 \otimes \sigma 2 \tag{4.7}$$

```
M > QTensors := [J0, J1, J2, J3];
```

$$QTensors := \left[\zeta 0 \otimes \sigma 0 + \zeta 1 \otimes \sigma 1 + \zeta 2 \otimes \sigma 2 + \zeta 3 \otimes \sigma 3, -\frac{R^4-a}{2R^3}\, \zeta 0 \otimes \sigma 3 \right. \tag{4.8}$$

$$+ \zeta1 \otimes \sigma2 - \zeta2 \otimes \sigma1 + \frac{2 R^3}{R^4 - a} \zeta3 \otimes \sigma0, \frac{\sqrt{R^4 - a}}{2 R} \zeta0 \otimes \sigma2$$

$$+ \frac{\sqrt{R^4 - a}}{R^2} \zeta1 \otimes \sigma3 - \frac{2 R}{\sqrt{R^4 - a}} \zeta2 \otimes \sigma0 - \frac{R^2}{\sqrt{R^4 - a}} \zeta3 \otimes \sigma1,$$

$$\frac{\sqrt{R^4 - a}}{2 R} \zeta0 \otimes \sigma1 - \frac{2 R}{\sqrt{R^4 - a}} \zeta1 \otimes \sigma0 - \frac{\sqrt{R^4 - a}}{R^2} \zeta2 \otimes \sigma3$$

$$+ \frac{R^2}{\sqrt{R^4 - a}} \zeta3 \otimes \sigma2 \Big]$$

Check that all our 1-1 tensors are parallel.

```
M > map(CovariantDerivative, QTensors, C);
```

$$[0\, \zeta0 \otimes \sigma0 \otimes \sigma0, 0\, \zeta0 \otimes \sigma0 \otimes \sigma0, 0\, \zeta0 \otimes \sigma0 \otimes \sigma0, 0\, \zeta0 \otimes \sigma0 \otimes \sigma0] \tag{4.9}$$

```
M > QMatrices := map(convert, map(convert, QTensors, DGArray),
    Matrix);
```

$$QMatrices := \left[\begin{bmatrix} 1 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 & -\frac{R^4 - a}{2 R^3} \\ 0 & 0 & 1 & 0 \\ 0 & -1 & 0 & 0 \\ \frac{2 R^3}{R^4 - a} & 0 & 0 & 0 \end{bmatrix}, \right. \tag{4.10}$$

$$\begin{bmatrix} 0 & 0 & \frac{\sqrt{R^4 - a}}{2 R} & 0 \\ 0 & 0 & 0 & \frac{\sqrt{R^4 - a}}{R^2} \\ -\frac{2 R}{\sqrt{R^4 - a}} & 0 & 0 & 0 \\ 0 & -\frac{R^2}{\sqrt{R^4 - a}} & 0 & 0 \end{bmatrix},$$

$$\left[\begin{bmatrix} 0 & \frac{\sqrt{R^4 - a}}{2R} & 0 & 0 \\ -\frac{2R}{\sqrt{R^4 - a}} & 0 & 0 & 0 \\ 0 & 0 & 0 & -\frac{\sqrt{R^4 - a}}{R^2} \\ 0 & 0 & \frac{R^2}{\sqrt{R^4 - a}} & 0 \end{bmatrix}\right]$$

Show that these 4 matrices define an algebra using ordinary matrix multiplication.

```
M > simplify(AlgebraData(QMatrices, proc(A,B) A.B end, alg))
    assuming R^4 - a >0;
```

$$[e1^2 = e1, e1 \cdot e2 = e2, e1 \cdot e3 = e3, e1 \cdot e4 = e4, e2 \cdot e1 = e2, e2^2 = -e1, e2 \cdot e3 = e4, e2 \cdot e4 = -e3, e3 \cdot e1 = e3, e3 \cdot e2 = -e4, e3^2 = -e1, e3 \cdot e4 = e2, e4 \cdot e1 = e4, e4 \cdot e2 = e3, e4 \cdot e3 = -e2, e4^2 = -e1] \quad (4.11)$$

The multiplication table matches exactly that of the quaternions!

```
M > AlgebraLibraryData("Quaternions", q);
```

$$[e1^2 = e1, e1 \cdot e2 = e2, e1 \cdot e3 = e3, e1 \cdot e4 = e4, e2 \cdot e1 = e2, e2^2 = -e1, e2 \cdot e3 = e4, e2 \cdot e4 = -e3, e3 \cdot e1 = e3, e3 \cdot e2 = -e4, e3^2 = -e1, e3 \cdot e4 = e2, e4 \cdot e1 = e4, e4 \cdot e2 = e3, e4 \cdot e3 = -e2, e4^2 = -e1] \quad (4.12)$$

Kahler Metric, Kahler Potential, and Holomorphic Sectional Curvature

In this section we shall analyze the complex structure defined by the 1-1 tensor J1. First we calculate the eigenforms for this tensor and use these to define a new frame.

```
M > lambda, P := DGEigenTensors(J1, [sigma0, sigma1, sigma2,
    sigma3]);
```

$$\lambda, P := \begin{bmatrix} I \\ I \\ -I \\ -I \end{bmatrix}, \left[-\frac{2\,I\,R^3}{R^4 - a}\,\sigma 0 + \sigma 3, I\,\sigma 1 + \sigma 2, \frac{2\,I\,R^3}{R^4 - a}\,\sigma 0 + \sigma 3, -I\,\sigma 1 + \sigma 2\right] \quad (5.1)$$

The 1st and 3rd forms are complex conjugates. The 2nd and 4th forms are complex conjugates.

```
N > DGconjugate(P[1]) &minus P[3];
```
$$0\ \sigma 0 \tag{5.2}$$

```
M > DGconjugate(P[2]) &minus P[4];
```
$$0\ \sigma 0 \tag{5.3}$$

```
M > FD2 := FrameData(P, N);
```
$$FD2 := \left[d\,\Theta 1 = \frac{\mathrm{I}}{2}\,\Theta 2 \wedge \Theta 4, d\,\Theta 2 = \frac{\mathrm{I}\,\Theta 1 \wedge \Theta 2}{2} - \frac{\mathrm{I}\,\Theta 2 \wedge \Theta 3}{2}, d\,\Theta 3 = \frac{\mathrm{I}}{2}\,\Theta 2 \wedge \Theta 4, d\,\Theta 4 = -\frac{\mathrm{I}\,\Theta 1 \wedge \Theta 4}{2} - \frac{\mathrm{I}\,\Theta 3 \wedge \Theta 4}{2}\right] \tag{5.4}$$

```
M > DGsetup(FD2, '[U1, U2, V1, V2]', '[omega1, omega2, theta1,
theta2]');
```
frame name: N (5.5)

Now we transform the metric and the complex structure J1 to the new frame.

```
N > IdMtoN := Transformation(M, N, [R = R, theta = theta, phi =
phi, psi = psi]);
```
$$IdMtoN := [R = R, \theta = \theta, \phi = \phi, \psi = \psi] \tag{5.6}$$

```
M > IdNtoM := Transformation(N, M, [R = R, theta = theta, phi =
phi, psi = psi]);
```
$$IdNtoM := [R = R, \theta = \theta, \phi = \phi, \psi = \psi] \tag{5.7}$$

Here in the Eguchi-Hanson metric in the new frame:

```
N > gN := Pullback(IdNtoM, g);
```
$$gN := \frac{R^4 - a}{8\,R^2}\,\omega 1 \otimes \theta 1 + \frac{R^2}{8}\,\omega 2 \otimes \theta 2 + \frac{R^4 - a}{8\,R^2}\,\theta 1 \otimes \omega 1 + \frac{R^2}{8}\,\theta 2 \otimes \omega 2 \tag{5.8}$$

```
N > JN := PushPullTensor(IdMtoN, IdNtoM, J1);
```
$$JN := \mathrm{I}\,U1 \otimes \omega 1 + \mathrm{I}\,U2 \otimes \omega 2 - \mathrm{I}\,V1 \otimes \theta 1 - \mathrm{I}\,V2 \otimes \theta 2 \tag{5.9}$$

Here is our complex structure.

```
N > JN := (1/I) &mult JN;
```
$$JN := U1 \otimes \omega 1 + U2 \otimes \omega 2 - V1 \otimes \theta 1 - V2 \otimes \theta 2 \tag{5.10}$$

Here is the Kahler form.

```
N > Omega := KahlerForm(gN, JN);
```
$$\Omega := \frac{R^4 - a}{8\,R^2}\,\omega 1 \wedge \theta 1 + \frac{R^2}{8}\,\omega 2 \wedge \theta 2 \tag{5.11}$$

This form is closed.

```
N > ExteriorDerivative(Omega);
```

$$0\,\omega 1 \wedge \omega 2 \wedge \theta 1 \tag{5.12}$$

Here is the Kahler potential for the metric.

```
N > rho := KahlerPotential(Omega, JN, ansatz = F(R));
```

$$\rho := 2 R^2 - 2\sqrt{a}\,\operatorname{arctanh}\left(\frac{R^2}{\sqrt{a}}\right) + \_C1 \tag{5.13}$$

```
N > alpha := DolbeaultExteriorDerivative(rho,[0,1], JN);
```

$$\alpha := \left(\frac{R^2}{2} + \frac{I\,R^2}{2}\right)\omega 1 + \left(\frac{R^2}{2} - \frac{I\,R^2}{2}\right)\theta 1 \tag{5.14}$$

```
N > beta := DolbeaultExteriorDerivative(alpha,[1, 0], JN);
```

$$\beta := -\frac{\frac{I}{4}\left(R^4 - a\right)}{R^2}\,\omega 1 \wedge \theta 1 - \frac{I}{4} R^2\,\omega 2 \wedge \theta 2 \tag{5.15}$$

```
N > evalDG(I/2 *beta - Omega);
```

$$0\,\omega 1 \wedge \omega 2 \tag{5.16}$$

Finally, here is the holomorphic sectional curvature:

```
N > C := CurvatureTensor(gN):
N > HolomorphicSectionalCurvature(gN, C, U1 + V1, JN);
```

$$-\frac{a\left(R^4 - a\right)}{R^8} \tag{5.17}$$

Complex Coordinates

```
N > ExteriorDifferentialSystems:-FirstIntegrals([omega1, omega2])
;
```

$$\left[I \ln\left(\frac{1 - \cos(\theta)}{\sin(\theta)}\right) + \phi, -I \ln(\sin(\theta)) - \frac{I \ln\left(R^4 - a\right)}{2} + \psi\right] \tag{6.1}$$

```
N > ExteriorDifferentialSystems:-FirstIntegrals([theta1, theta2])
;
```

$$\left[-I \ln\left(\frac{1 - \cos(\theta)}{\sin(\theta)}\right) + \phi, I \ln(\sin(\theta)) + \frac{I \ln\left(R^4 - a\right)}{2} + \psi\right] \tag{6.2}$$